**Modeling Categorical Outcomes via SAS GLIMMIX and STATA MEOLOGIT/MLOGIT**
***(data, syntax, and output available for SAS and STATA electronically)***

The (likely fake) data for this example came from: http://www.ats.ucla.edu/stat/sas/dae/ologit.htm.
In this example we will predict a categorical decision to apply to grad school (no, maybe, yes) based on student GPA, whether one of their parents has a graduate degree, and whether they attended private school.

## STATA Syntax and Output for Data Manipulation:

```
* Import data, transform variables, apply value formats
use "$filesave\ologit.dta", clear
gen parentgd=pared
gen gpa3=gpa-3
gen private=0
recode private (0=1) if public==0
label variable apply    "apply: 0=No, 1=Maybe, 2=Yes"
label variable parentgd "parentgd: Parent Has Graduate Degree (0=N,1=Y)"
label variable private  "private: Student Attends Private University (0=N,1=Y)"
label variable gpa3     "gpa3: Student GPA (0=3)"
label define f3apply     0 "0No" 1 "1Maybe" 2 "2Yes"
label values apply f3apply
```

## SAS Syntax and Output for Data Manipulation:

```
* Creating formats for categorical variables;
PROC FORMAT; VALUE F3apply 0="0No" 1="1Maybe" 2="2Yes"; RUN;

* Import data into work library, transform variables, apply value formats;
DATA work.ologit; SET filesave.ologit;
    parentGD=pared;
    GPA3=GPA-3;
         IF public=1 THEN private=0;
    ELSE IF public=0 THEN private=1;
    ELSE IF public=. THEN private=.;
    LABEL apply=    "apply: 0=No, 1=Maybe, 2=Yes"
          parentGD= "parentGD: Parent Has Graduate Degree (0=N,1=Y)"
          private=  "private: Student Attends Private University (0=N,1=Y)"
          GPA3=     "GPA3: Student GPA (0=3)";
       FORMAT apply F3apply.;
RUN;

TITLE1 "DESCRIPTIVES FOR STUDY VARIABLES";
PROC MEANS DATA=work.ologit; VAR GPA; RUN;
PROC FREQ DATA=work.ologit; TABLE apply3 parentGD private; RUN; TITLE1;
```

apply: 0=No, 1=Maybe, 2=Yes

| APPLY | Frequency | Percent | Cumulative Frequency | Cumulative Percent |
|---|---|---|---|---|
| 0No | 220 | 55.00 | 220 | 55.00 |
| 1Maybe | 140 | 35.00 | 360 | 90.00 |
| 2Yes | 40 | 10.00 | 400 | 100.00 |

**Empty Ordinal Model predicting the cumulative logit of 3-category apply:**

$Logit(Apply_i > 0) = \beta_{01}$ → $Probability(Apply_i > 0) = \frac{exp(\beta_{01})}{1+exp(\beta_{01})}$

$Logit(Apply_i > 1) = \beta_{02}$ → $Probability(Apply_i > 1) = \frac{exp(\beta_{02})}{1+exp(\beta_{02})}$

```
display as result "EMPTY ORDINAL MODEL PREDICTING CATEGORICAL DV"
display as result "INTERCEPTS ARE BACKWARDS (GIVE LOGIT OF LOWER CATEGORY)"
meologit apply ,
estat ic, n(400),
nlcom 1/(1+exp(-1*(_b[cut1:_cons]))) // intercept for y<1 in probability
nlcom 1/(1+exp(-1*(_b[cut2:_cons]))) // intercept for y<2 in probability
```

```
TITLE1 "EMPTY ORDINAL MODEL PREDICTING CATEGORICAL DV";
PROC GLIMMIX DATA=work.ologit NOCLPRINT NOITPRINT GRADIENT METHOD=QUAD;
MODEL apply (DESCENDING) = / SOLUTION LINK=CLOGIT DIST=MULT;
* ILINK requests logit estimate to be transformed into probability, but does not give each;
ESTIMATE "Intercept" intercept 1 / ILINK;
RUN; TITLE1;
```

**SAS Output:**

```
      Convergence criterion (ABSGCONV=0.00001) satisfied.
```

Hooray! Our estimates are usable!

```
          Fit Statistics
-2 Log Likelihood             741.21
AIC  (smaller is better)      745.21
AICC (smaller is better)      745.24
BIC  (smaller is better)      753.19
CAIC (smaller is better)      755.19
HQIC (smaller is better)      748.37
```

Parameter Estimates

| Effect | apply: 0=No, 1=Maybe, 2=Yes | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Gradient |
|---|---|---|---|---|---|---|---|
| Intercept | 2Yes | -2.1972 | 0.1667 | 398 | -13.18 | <.0001 | 6.19E-15 |
| Intercept | 1Maybe | -0.2007 | 0.1005 | 398 | -2.00 | 0.0465 | 3.66E-15 |

Estimates

Estimate = predicted logit

Mean = probability

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Mean | Standard Error Mean |
|---|---|---|---|---|---|---|---|
| Intercept | -2.1972 | 0.1667 | 398 | -13.18 | <.0001 | 0.1000 | 0.01500 |

$$Probability(Apply_i > 0) = \frac{exp(-0.2007)}{[1+exp(-0.2007)]} = .450$$

$$Probability(Apply_i > 1) = \frac{exp(-2.1972)}{[1+exp(-2.1972)]} = .100$$

apply: 0=Not, 1=Maybe, 2=Yes

| APPLY | Frequency | Percent | Cumulative Frequency | Cumulative Percent |
|---|---|---|---|---|
| 0No | 220 | 55.00 | 220 | 55.00 |
| 1Maybe | 140 | 35.00 | 360 | 90.00 |
| 2Yes | 40 | 10.00 | 400 | 100.00 |

**Let's add some predictors...**

$$Logit(Apply_i > 0) = \beta_{01} + \beta_1(GPA_i - 3) + \beta_2(ParentGD_i) + \beta_3(Private_i)$$
$$Logit(Apply_i > 1) = \beta_{02} + \beta_1(GPA_i - 3) + \beta_2(ParentGD_i) + \beta_3(Private_i)$$

```
display as result "MAIN EFFECTS ORDINAL MODEL PREDICTING CATEGORICAL DV"
display as result "INTERCEPTS ARE BACKWARDS (GIVE LOGIT OF LOWER CATEGORY)"
melogit apply c.gpa3 c.parentgd c.private,
estat ic, n(400),
margins, at(c.gpa3=(-1(1)1) c.parentgd=(0(1)1) c.private=(0(1)1)) predict(xb) // logits
margins, at(c.gpa3=(-1(1)1) c.parentgd=(0(1)1) c.private=(0(1)1))             // probabilities
```

```
TITLE1 "MAIN EFFECTS ORDINAL MODEL PREDICTING CATEGORICAL DV";
PROC GLIMMIX DATA=work.ologit NOCLPRINT NOITPRINT GRADIENT METHOD=QUAD;
MODEL apply (DESCENDING) = GPA3 parentGD private / SOLUTION LINK=CLOGIT DIST=MULT ODDSRATIO(AT
GPA3=0);
ESTIMATE "Intercept for No  Degree, Public,  GPA=2" int 1 GPA3 -1 parentGD 0 private 0 / ILINK;
ESTIMATE "Intercept for No  Degree, Private, GPA=2" int 1 GPA3 -1 parentGD 0 private 1 / ILINK;
ESTIMATE "Intercept for Yes Degree, Public,  GPA=2" int 1 GPA3 -1 parentGD 1 private 0 / ILINK;
ESTIMATE "Intercept for Yes Degree, Private, GPA=2" int 1 GPA3 -1 parentGD 1 private 1 / ILINK;
ESTIMATE "Intercept for No  Degree, Public,  GPA=3" int 1 GPA3  0 parentGD 0 private 0 / ILINK;
ESTIMATE "Intercept for No  Degree, Private, GPA=3" int 1 GPA3  0 parentGD 0 private 1 / ILINK;
ESTIMATE "Intercept for Yes Degree, Public,  GPA=3" int 1 GPA3  0 parentGD 1 private 0 / ILINK;
ESTIMATE "Intercept for Yes Degree, Private, GPA=3" int 1 GPA3  0 parentGD 1 private 1 / ILINK;
ESTIMATE "Intercept for No  Degree, Public,  GPA=4" int 1 GPA3  1 parentGD 0 private 0 / ILINK;
ESTIMATE "Intercept for No  Degree, Private, GPA=4" int 1 GPA3  1 parentGD 0 private 1 / ILINK;
ESTIMATE "Intercept for Yes Degree, Public,  GPA=4" int 1 GPA3  1 parentGD 1 private 0 / ILINK;
ESTIMATE "Intercept for Yes Degree, Private, GPA=4" int 1 GPA3  1 parentGD 1 private 1 / ILINK;
ESTIMATE "Slope for GPA" GPA3 1 / ILINK; * Example of non-sense ILINK for a slope;
RUN; TITLE1;
```

**SAS Output:**

```
          Fit Statistics
-2 Log Likelihood            717.02
AIC  (smaller is better)     727.02
BIC  (smaller is better)     746.98
```

```
                                  Parameter Estimates
            apply:
            0=No,
            1=Maybe,                 Standard
Effect      2=Yes       Estimate        Error       DF    t Value    Pr > |t|    Gradient
Intercept   2Yes          -2.5102       0.3192      395      -7.86      <.0001    7.07E-12
Intercept   1Maybe        -0.4148       0.2830      395      -1.47      0.1435    -262E-13
GPA3                       0.6157       0.2606      395       2.36      0.0186    5.41E-13
parentGD                   1.0477       0.2658      395       3.94      <.0001    1.92E-12
private                   0.05868       0.2979      395       0.20      0.8439    6.07E-12
```

**Interpret each main effect…**

**Intercept for 1Maybe:**

**Intercept for 2Yes:**

**GPA3:**

**parentGD:**

**private:**

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Mean | Standard Error Mean |
|---|---|---|---|---|---|---|---|
| Intercept for No Degree, Public, GPA=2 | -3.1260 | 0.4525 | 395 | -6.91 | <.0001 | 0.04205 | 0.01823 |
| Intercept for No Degree, Private, GPA=2 | -3.0673 | 0.3211 | 395 | -9.55 | <.0001 | 0.04448 | 0.01365 |
| Intercept for Yes Degree, Public, GPA=2 | -2.0783 | 0.4971 | 395 | -4.18 | <.0001 | 0.1112 | 0.04914 |
| Intercept for Yes Degree, Private, GPA=2 | -2.0196 | 0.3893 | 395 | -5.19 | <.0001 | 0.1172 | 0.04027 |
| **Intercept for No Degree, Public, GPA=3** | **-2.5102** | **0.3192** | **395** | **-7.86** | **<.0001** | **0.07515** | **0.02218** |
| Intercept for No Degree, Private, GPA=3 | -2.4515 | 0.1870 | 395 | -13.11 | <.0001 | 0.07933 | 0.01366 |
| Intercept for Yes Degree, Public, GPA=3 | -1.4625 | 0.3608 | 395 | -4.05 | <.0001 | 0.1881 | 0.05510 |
| Intercept for Yes Degree, Private, GPA=3 | -1.4039 | 0.2634 | 395 | -5.33 | <.0001 | 0.1972 | 0.04170 |
| Intercept for No Degree, Public, GPA=4 | -1.8945 | 0.3672 | 395 | -5.16 | <.0001 | 0.1307 | 0.04173 |
| Intercept for No Degree, Private, GPA=4 | -1.8358 | 0.3204 | 395 | -5.73 | <.0001 | 0.1376 | 0.03801 |
| Intercept for Yes Degree, Public, GPA=4 | -0.8468 | 0.3862 | 395 | -2.19 | 0.0289 | 0.3001 | 0.08111 |
| Intercept for Yes Degree, Private, GPA=4 | -0.7881 | 0.3508 | 395 | -2.25 | 0.0252 | 0.3126 | 0.07538 |
| Slope for GPA | 0.6157 | 0.2606 | 395 | 2.36 | 0.0186 | 0.6493 | 0.05935 |

The last line illustrates why you cannot "un-logit" a slope… the difference between the intercepts per unit GPA in logits is a constant 0.6157, but the difference in probability is not constant (and is not 0). Similarly, the difference between the groups is constant in logits, but is NOT constant in probability (it depends where you are on the probability scale, which depends upon the submodel and the values of the other predictors).

**These models have made an important (but empirically testable) assumption of proportional odds: that all predictor slopes are equal across sub-models. To get a sense of what the slopes would be for each sub-model otherwise we can take advantage of the fact that our outcome only has 3 categories and fit it using a nominal (generalized logit) model instead, with "1maybe" as the reference outcome.**

**However, STATA does not appear to have a mixed effects version of this model, so we will use a univariate version instead (although it is supposed to be possible to do using GSEM).**

---

**Empty Nominal Model predicting the generalized logit of 3-category apply (like dummy-coding):**

$$Logit(Apply_i = 0\ instead\ of\ 1) = \beta_{00} \rightarrow Probability(Apply_i = 0\ instead\ of\ 1) = \frac{exp(\beta_{00})}{1+exp(\beta_{00})}$$

$$Logit(Apply_i = 2\ instead\ of\ 1) = \beta_{02} \rightarrow Probability(Apply_i = 2\ instead\ of\ 1) = \frac{exp(\beta_{02})}{1+exp(\beta_{02})}$$

```
display as result "EMPTY ORDINAL MODEL PREDICTING CATEGORICAL DV"
display as result "INTERCEPTS ARE NO LONGER BACKWARDS (GIVES LOGIT OF NON-BASE CATEGORY)"
mlogit apply ,  baseoutcome(1),
estat ic, n(400),
nlcom 1/(1+exp(-1*(_b[0No:_cons])))    // intercept for y=0 instead of 1 in probability
nlcom 1/(1+exp(-1*(_b[2Maybe:_cons]))) // intercept for y=2 instead of 1 in probability
```

```
TITLE1 "EMPTY NOMINAL MODEL PREDICTING CATEGORICAL DV";
PROC GLIMMIX DATA=work.ologit NOCLPRINT NOITPRINT GRADIENT METHOD=QUAD;
MODEL apply (REFERENCE="1Maybe") = / SOLUTION LINK=GLOGIT DIST=MULT;
* ILINK requests logit estimate to be transformed into probability;
ESTIMATE "Intercept" intercept 1 / ILINK BYCAT; * BYCAT gives per submodel;
RUN; TITLE1;
```

**SAS Output:**

Fit Statistics

| | |
|---|---|
| -2 Log Likelihood | 741.21 |
| AIC (smaller is better) | 745.21 |
| BIC (smaller is better) | 753.19 |

Parameter Estimates

| Effect | apply: 0=No, 1=Maybe, 2=Yes | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Gradient |
|---|---|---|---|---|---|---|---|
| Intercept | 0No | 0.4520 | 0.1081 | 398 | 4.18 | <.0001 | -107E-16 |
| Intercept | 2Yes | -1.2528 | 0.1793 | 398 | -6.99 | <.0001 | 1.35E-14 |

Estimates

| Label | apply: 0=No, 1=Maybe, 2=Yes | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Mean | Standard Error Mean |
|---|---|---|---|---|---|---|---|---|
| Intercept | 0No | 0.4520 | 0.1081 | 398 | 4.18 | <.0001 | 0.5500 | 0.02487 |
| Intercept | 2Yes | -1.2528 | 0.1793 | 398 | -6.99 | <.0001 | 0.1000 | 0.01500 |

$$Given\ that\ y = 0\ or\ y = 1: \quad Probability(Apply_i = 0) = \frac{exp(0.4520)}{[1 + exp(0.4520)]} / = .6111$$

$$Given\ that\ y = 0\ or\ y = 1: \quad Probability(Apply_i = 2) = \frac{exp(-1.2528)}{[1 + exp(-1.2528)]} / = .2222$$

apply: 0=Not, 1=Maybe, 2=Yes

| APPLY | Frequency | Percent | Cumulative Frequency | Cumulative Percent |
|---|---|---|---|---|
| 0No | 220 | 55.00 | 220 | 55.00 |
| 1Maybe | 140 | 35.00 | 360 | 90.00 |
| 2Yes | 40 | 10.00 | 400 | 100.00 |

Probability that y=0 or y=1: .90, so y=0 is .55/.90 = .6111
Probability that y=2 or y=1: .45, so y=2 is .10/.45 = .2222

**The logits translate into conditional probabilities, the predicted "mean" is the marginal probability.**

---

**Let's add some predictors...**

$$Logit(Apply_i = 0\ instead\ of\ 1) = \beta_{00} + \beta_{10}(GPA_i - 3) + \beta_{20}(ParentGD_i) + \beta_{30}(Private_i)$$
$$Logit(Apply_i = 2\ instead\ of\ 1) = \beta_{02} + \beta_{12}(GPA_i - 3) + \beta_{22}(ParentGD_i) + \beta_{32}(Private_i)$$

```
display as result "MAIN EFFECTS ORDINAL MODEL PREDICTING CATEGORICAL DV"
display as result "INTERCEPTS ARE NO LONGER BACKWARDS (GIVES LOGIT OF NON-BASE CATEGORY)"
mlogit apply c.gpa3 c.parentgd c.private, baseoutcome(1),
estat ic, n(400),
margins, at(c.gpa3=(-1(1)1) c.parentgd=(0(1)1) c.private=(0(1)1)) predict(xb) // logits
margins, at(c.gpa3=(-1(1)1) c.parentgd=(0(1)1) c.private=(0(1)1))             // probabilities

TITLE1 "MAIN EFFECTS NOMINAL MODEL PREDICTING CATEGORICAL DV";
PROC GLIMMIX DATA=work.ologit NOCLPRINT NOITPRINT GRADIENT METHOD=QUAD;
MODEL apply (REFERENCE="1Maybe") = GPA3 parentGD private / SOLUTION LINK=GLOGIT DIST=MULT
ODDSRATIO(AT GPA3=0);
ESTIMATE "Int for No  Degree, Public,  GPA=2" int 1 GPA3 -1 parentGD 0 private 0 / ILINK BYCAT;
ESTIMATE "Int for No  Degree, Private, GPA=2" int 1 GPA3 -1 parentGD 0 private 1 / ILINK BYCAT;
ESTIMATE "Int for Yes Degree, Public,  GPA=2" int 1 GPA3 -1 parentGD 1 private 0 / ILINK BYCAT;
ESTIMATE "Int for Yes Degree, Private, GPA=2" int 1 GPA3 -1 parentGD 1 private 1 / ILINK BYCAT;
ESTIMATE "Int for No  Degree, Public,  GPA=3" int 1 GPA3  0 parentGD 0 private 0 / ILINK BYCAT;
ESTIMATE "Int for No  Degree, Private, GPA=3" int 1 GPA3  0 parentGD 0 private 1 / ILINK BYCAT;
ESTIMATE "Int for Yes Degree, Public,  GPA=3" int 1 GPA3  0 parentGD 1 private 0 / ILINK BYCAT;
ESTIMATE "Int for Yes Degree, Private, GPA=3" int 1 GPA3  0 parentGD 1 private 1 / ILINK BYCAT;
ESTIMATE "Int for No  Degree, Public,  GPA=4" int 1 GPA3  1 parentGD 0 private 0 / ILINK BYCAT;
ESTIMATE "Int for No  Degree, Private, GPA=4" int 1 GPA3  1 parentGD 0 private 1 / ILINK BYCAT;
ESTIMATE "Int for Yes Degree, Public,  GPA=4" int 1 GPA3  1 parentGD 1 private 0 / ILINK BYCAT;
ESTIMATE "Int for Yes Degree, Private, GPA=4" int 1 GPA3  1 parentGD 1 private 1 / ILINK BYCAT;
RUN; TITLE1;
```

## SAS Output:

| Fit Statistics | |
|---|---|
| -2 Log Likelihood | 713.99 |
| AIC (smaller is better) | 729.99 |
| BIC (smaller is better) | 761.93 |

Type III Tests of Fixed Effects

| Effect | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| GPA3 | 2 | 392 | 2.48 | 0.0853 |
| parentGD | 2 | 392 | 6.93 | 0.0011 |
| private | 2 | 392 | 1.52 | 0.2204 |

Parameter Estimates

| Effect | apply: 0=No, 1=Maybe, 2=Yes | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Gradient |
|---|---|---|---|---|---|---|---|
| Intercept | 0No | 0.9515 | 0.3258 | 392 | 2.92 | 0.0037 | -2.37E-8 |
| Intercept | 2Yes | -0.7641 | 0.4511 | 392 | -1.69 | 0.0911 | 3.684E-8 |
| GPA3 | 0No | -0.4488 | 0.2902 | 392 | -1.55 | 0.1228 | 8.02E-11 |
| GPA3 | 2Yes | 0.4753 | 0.4871 | 392 | 0.98 | 0.3298 | -597E-12 |
| **parentGD** | **0No** | **-0.9516** | **0.3171** | **392** | **-3.00** | **0.0029** | **-1.6E-8** |
| parentGD | 2Yes | 0.4225 | 0.4083 | 392 | 1.03 | 0.3014 | 2.634E-8 |
| private | 0No | -0.4188 | 0.3433 | 392 | -1.22 | 0.2232 | -1.27E-8 |
| **private** | **2Yes** | **-0.7789** | **0.4706** | **392** | **-1.66** | **0.0987** | **2.266E-8** |

**It looks like parent graduate degree (no vs. yes) has a stronger effect on no vs. maybe, whereas public vs. private school has a stronger effect on maybe vs. yes. But are these effect sizes really different?**



**Estimates (how plots were made)**

| Label | apply: 0=No, 1=Maybe, 2=Yes | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Mean | Standard Error Mean |
|---|---|---|---|---|---|---|---|---|
| Int for No Degree, Public, GPA=2 | 0No | 1.4003 | 0.4714 | 392 | 2.97 | 0.0032 | 0.7588 | 0.07966 |
| Int for No Degree, Public, GPA=2 | 2Yes | -1.2393 | 0.7516 | 392 | -1.65 | 0.0999 | 0.05417 | 0.03582 |
| Int for No Degree, Private, GPA=2 | 0No | 0.9815 | 0.3014 | 392 | 3.26 | 0.0012 | 0.7020 | 0.05998 |
| Int for No Degree, Private, GPA=2 | 2Yes | -2.0182 | 0.5496 | 392 | -3.67 | 0.0003 | 0.03496 | 0.01755 |
| Int for Yes Degree, Public, GPA=2 | 0No | 0.4486 | 0.5564 | 392 | 0.81 | 0.4206 | 0.5207 | 0.1294 |
| Int for Yes Degree, Public, GPA=2 | 2Yes | -0.8168 | 0.8380 | 392 | -0.97 | 0.3303 | 0.1469 | 0.09749 |
| Int for Yes Degree, Private, GPA=2 | 0No | 0.02981 | 0.4335 | 392 | 0.07 | 0.9452 | 0.4614 | 0.1018 |
| Int for Yes Degree, Private, GPA=2 | 2Yes | -1.5957 | 0.6744 | 392 | -2.37 | 0.0185 | 0.09080 | 0.05246 |
| Int for No Degree, Public, GPA=3 | 0No | 0.9515 | 0.3258 | 392 | 2.92 | 0.0037 | 0.6386 | 0.06687 |
| Int for No Degree, Public, GPA=3 | 2Yes | -0.7641 | 0.4511 | 392 | -1.69 | 0.0911 | 0.1149 | 0.04087 |
| Int for No Degree, Private, GPA=3 | 0No | 0.5327 | 0.1261 | 392 | 4.23 | <.0001 | 0.5839 | 0.02887 |
| Int for No Degree, Private, GPA=3 | 2Yes | -1.5429 | 0.2322 | 392 | -6.64 | <.0001 | 0.07327 | 0.01487 |
| Int for Yes Degree, Public, GPA=3 | 0No | -0.00012 | 0.4111 | 392 | -0.00 | 0.9998 | 0.3689 | 0.08788 |
| Int for Yes Degree, Public, GPA=3 | 2Yes | -0.3416 | 0.5184 | 392 | -0.66 | 0.5104 | 0.2622 | 0.09139 |
| Int for Yes Degree, Private, GPA=3 | 0No | -0.4189 | 0.2974 | 392 | -1.41 | 0.1597 | 0.3315 | 0.06217 |
| Int for Yes Degree, Private, GPA=3 | 2Yes | -1.1204 | 0.3669 | 392 | -3.05 | 0.0024 | 0.1644 | 0.04752 |
| Int for No Degree, Public, GPA=4 | 0No | 0.5028 | 0.3982 | 392 | 1.26 | 0.2074 | 0.4859 | 0.09194 |
| Int for No Degree, Public, GPA=4 | 2Yes | -0.2888 | 0.5628 | 392 | -0.51 | 0.6082 | 0.2202 | 0.08758 |
| Int for No Degree, Private, GPA=4 | 0No | 0.08396 | 0.3307 | 392 | 0.25 | 0.7997 | 0.4473 | 0.07667 |
| Int for No Degree, Private, GPA=4 | 2Yes | -1.0677 | 0.5295 | 392 | -2.02 | 0.0444 | 0.1414 | 0.06020 |
| Int for Yes Degree, Public, GPA=4 | 0No | -0.4489 | 0.4438 | 392 | -1.01 | 0.3124 | 0.2295 | 0.07504 |
| Int for Yes Degree, Public, GPA=4 | 2Yes | 0.1337 | 0.5567 | 392 | 0.24 | 0.8103 | 0.4110 | 0.1242 |
| Int for Yes Degree, Private, GPA=4 | 0No | -0.8677 | 0.3967 | 392 | -2.19 | 0.0293 | 0.2160 | 0.06392 |
| Int for Yes Degree, Private, GPA=4 | 2Yes | -0.6451 | 0.5375 | 392 | -1.20 | 0.2308 | 0.2698 | 0.1005 |

---

**One way to test differences in effect size across sub-models is to stack the possible sub-model outcomes into a single outcome and run a multivariate model, which can approximate a nominal model, but allows explicit contrasts of these separate slopes. Here is the SAS code to do so:**

```
***** TRICKING A NOMINAL MODEL OUT OF A MULTIVARIATE MODEL;
***** provides a way to directly test differences between the sub-model slopes;

* Stack data into multivariate to get tests of proportional odds;
DATA work.ologitstacked; SET work.ologit;
IF apply=0 THEN DO; MaybevsNo=1; MaybevsYes=.; END;
IF apply=1 THEN DO; MaybevsNo=0; MaybevsYes=0; END;
IF apply=2 THEN DO; MaybevsNo=.; MaybevsYes=1; END;
DV10=1; DV12=0; y=MaybevsNo;  OUTPUT;
DV10=0; DV12=1; y=MaybevsYes; OUTPUT;
LABEL DV10= "DV10: Is testing Maybe vs. No  (0=N,1=Y)"
      DV12= "DV12: Is testing Maybe vs. Yes (0=N,1=Y)"
         y=  "y: Actual response"; RUN;

TITLE1 "EMPTY MULTIVARIATE MODEL PREDICTING DUMMY-CODED DVs";
PROC GLIMMIX DATA=work.ologitstacked NOCLPRINT NOITPRINT GRADIENT METHOD=QUAD;
MODEL y (DESCENDING) = DV10 DV12 / NOINT SOLUTION LINK=LOGIT DIST=BINARY;
ESTIMATE "Intercept for DV10" DV10 1 / ILINK;
ESTIMATE "Intercept for DV12" DV12 1 / ILINK;
RUN; TITLE1;
```

Parameter Estimates

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Gradient |
|---|---|---|---|---|---|---|
| DV10 | 0.4520 | 0.1081 | 538 | 4.18 | <.0001 | -125E-15 |
| DV12 | -1.2528 | 0.1793 | 538 | -6.99 | <.0001 | 4.36E-11 |

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Mean | Standard Error Mean |
|---|---|---|---|---|---|---|---|
| Intercept for DV10 | 0.4520 | 0.1081 | 538 | 4.18 | <.0001 | 0.6111 | 0.02569 |
| Intercept for DV12 | -1.2528 | 0.1793 | 538 | -6.99 | <.0001 | 0.2222 | 0.03099 |

**The model directly predicts the logit of 0 instead of 1 for DV10, and the logit of 2 instead of 1 for DV12, same as the nominal model (but note that conditional probabilities are given for the "mean").**

```
TITLE1 "MAIN EFFECTS MULTIVARIATE MODEL PREDICTING DUMMY-CODED DVs";
TITLE2 "TO GET DIRECT TESTS OF PROPORTIONAL ODDS";
PROC GLIMMIX DATA=work.ologitstacked NOCLPRINT NOITPRINT GRADIENT METHOD=QUAD;
MODEL y (DESCENDING) = DV10 DV12 DV10*GPA3 DV12*GPA3 DV10*parentGD DV12*parentGD
                       DV10*private DV12*private / NOINT SOLUTION LINK=LOGIT DIST=BINARY;
* Note that estimates below flip the sign of the DV10 effects (to test diffs in absolute value);
ESTIMATE "Difference in GPA Slope across Submodels"         DV10*GPA3     1  DV12*GPA3     1;
ESTIMATE "Difference in ParentGD Slope across Submodels"    DV10*parentGD 1  DV12*parentGD 1;
ESTIMATE "Difference in Private Slope across Submodels"     DV10*private  1  DV12*private  1;
RUN; TITLE1;
```

Parameter Estimates

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Gradient |
|---|---|---|---|---|---|---|
| DV10 | 0.9992 | 0.3319 | 532 | 3.01 | 0.0027 | -269E-13 |
| DV12 | -0.7699 | 0.4518 | 532 | -1.70 | 0.0889 | 2.31E-12 |
| DV10*GPA3 | -0.4828 | 0.2875 | 532 | -1.68 | 0.0937 | 9.11E-12 |
| DV12*GPA3 | 0.4813 | 0.5036 | 532 | 0.96 | 0.3397 | 6.72E-13 |
| **DV10*parentGD** | **-0.9833** | **0.3168** | **532** | **-3.10** | **0.0020** | **1.71E-12** |
| DV12*parentGD | 0.4992 | 0.4061 | 532 | 1.23 | 0.2195 | 1.43E-12 |
| DV10*private | -0.4682 | 0.3481 | 532 | -1.35 | 0.1792 | -118E-13 |
| **DV12*private** | **-0.7978** | **0.4786** | **532** | **-1.67** | **0.0961** | **2.22E-12** |

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Difference in GPA Slope across Submodels | -0.00150 | 0.5799 | 532 | -0.00 | 0.9979 |
| Difference in ParentGD Slope across Submodels | -0.4841 | 0.5151 | 532 | -0.94 | 0.3477 |
| Difference in Private Slope across Submodels | -1.2660 | 0.5918 | 532 | -2.14 | 0.0329 |

**Previous results from nominal model—the multivariate model results are similar but not the same:**

| Effect | apply: 0=No, 1=Maybe, 2=Yes | Estimate | Standard Error | DF | t Value | Pr > \|t\| | Gradient |
|---|---|---|---|---|---|---|---|
| Intercept | 0No | 0.9515 | 0.3258 | 392 | 2.92 | 0.0037 | -2.37E-8 |
| Intercept | 2Yes | -0.7641 | 0.4511 | 392 | -1.69 | 0.0911 | 3.684E-8 |
| GPA3 | 0No | -0.4488 | 0.2902 | 392 | -1.55 | 0.1228 | 8.02E-11 |
| GPA3 | 2Yes | 0.4753 | 0.4871 | 392 | 0.98 | 0.3298 | -597E-12 |
| **parentGD** | **0No** | **-0.9516** | **0.3171** | **392** | **-3.00** | **0.0029** | **-1.6E-8** |
| parentGD | 2Yes | 0.4225 | 0.4083 | 392 | 1.03 | 0.3014 | 2.634E-8 |
| private | 0No | -0.4188 | 0.3433 | 392 | -1.22 | 0.2232 | -1.27E-8 |
| **private** | **2Yes** | **-0.7789** | **0.4706** | **392** | **-1.66** | **0.0987** | **2.266E-8** |